```
(*
equations24to27.nb

 produced by D.F.Gochberg for paper:
  D.F.Gochberg, M.D.Does, Z.Zu, C.L.Lankford. Towards
  an analytic solution for pulsed CEST. NMR in Biomedicine 2017

  You are free to use this code for non-commercial purposes,
but please cite the above manuscript if you use the code,or parts thereof,
to help produce a manuscript or presentation figure,as appropriate.Thanks.

  Based on: analytic_CEST_short_times2_keep_w1.nb and lab notebook 27,
p.25 (Dan's reminder to himself)
*)

(* Start with coupled Bloch eqns.  "a" pool is
 constant at short times.  Solve resulting eqn using Laplace
 transforms and following approach of Morris and Chilvers,
General analytic solutions of the Bloch equations.  JMR series A,
107, 236-238 (1994): *)

(* from Eqn A1: *)
```

$$\text{laplaceA} = \begin{pmatrix} p + r2b + kb & -dwb & 0 \\ dwb & p + r2b + kb & -w1 \\ 0 & w1 & p + r1b + kb \end{pmatrix};$$

$$\text{laplaceB} = \begin{pmatrix} kb * dwa * w1 * zai / (p * (w1^2 + dwa^2)) \\ 0 \\ (((kb * dwa^2 * zai) / (w1^2 + dwa^2) + r1b) / p + zbi) \end{pmatrix};$$

```
(* solve for transformed b-pool: *)
laplaceuvMbz = Inverse[laplaceA].laplaceB;
MatrixForm[laplaceuvMbz]
```

$$\begin{pmatrix} \frac{dwa\ kb\ w1\ ((kb+p+r1b)\ (kb+p+r2b)+w1^2)\ zai}{p\ (dwa^2+w1^2)\ ((kb+p+r1b)\ (dwb^2+(kb+p+r2b)^2)+(kb+p+r2b)\ w1^2)} + \frac{dwb\ w1\ \left(\frac{r1b+\frac{dwa^2\ kb\ zai}{dwa^2+w1^2}}{p}+zbi\right)}{(kb+p+r1b)\ (dwb^2+(kb+p+r2b)^2)+(kb+p+r2b)\ w1^2} \\ \frac{dwa\ kb\ (-dwb\ kb-dwb\ p-dwb\ r1b)\ w1\ zai}{p\ (dwa^2+w1^2)\ ((kb+p+r1b)\ (dwb^2+(kb+p+r2b)^2)+(kb+p+r2b)\ w1^2)} + \frac{(kb+p+r2b)\ w1\ \left(\frac{r1b+\frac{dwa^2\ kb\ zai}{dwa^2+w1^2}}{p}+zbi\right)}{(kb+p+r1b)\ (dwb^2+(kb+p+r2b)^2)+(kb+p+r2b)\ w1^2} \\ \frac{dwa\ dwb\ kb\ w1^2\ zai}{p\ (dwa^2+w1^2)\ ((kb+p+r1b)\ (dwb^2+(kb+p+r2b)^2)+(kb+p+r2b)\ w1^2)} + \frac{(dwb^2+(kb+p+r2b)^2)\ \left(\frac{r1b+\frac{dwa^2\ kb\ zai}{dwa^2+w1^2}}{p}+zbi\right)}{(kb+p+r1b)\ (dwb^2+(kb+p+r2b)^2)+(kb+p+r2b)\ w1^2} \end{pmatrix}$$

```
(* get Laplace transform of Zb: *)
laplaceMbz =
 Collect[laplaceuvMbz[[3, 1]], (kb + p + r1b) (dwb^2 + (kb + p + r2b)^2) + (kb + p + r2b) w1^2]
```

$$\frac{\frac{\text{dwa dwb kb w1}^2\text{ zai}}{\text{p}\left(\text{dwa}^2+\text{w1}^2\right)}+\left(\text{dwb}^2+(\text{kb}+\text{p}+\text{r2b})^2\right)\left(\frac{\text{r1b}+\frac{\text{dwa}^2\text{ kb zai}}{\text{dwa}^2+\text{w1}^2}}{\text{p}}+\text{zbi}\right)}{(\text{kb}+\text{p}+\text{r1b})\left(\text{dwb}^2+(\text{kb}+\text{p}+\text{r2b})^2\right)+(\text{kb}+\text{p}+\text{r2b})\text{ w1}^2}$$

```
(* Like Morris & Chivers, with alpha→alpha+k, beta→beta+k in a and b*)

(*
Now let's do partial fractions approach.  Goal is to
  get lapaceMz in form aA/(p+a) + (bB*(p+b)+cC)/((p+b)^2+s^2) +
 dD/p.  Here "aA" is an amplitude with rate "a",
and so on.  For correspondence to the paper:
   aA → cb_R1p,fast, a→ R1p, fast,
bB → cb_cos, w_eff,b, b → R2p,b,
cC/s → cb_sin, w_eff,b, s→ w_eff,b

 I will not solve for a, b, and s,
but will instead use the Torry approximations (alpha→alpha+k, beta→beta+k)
 in a separate file to derive eqn 28.  These rate values will be
 referred to aNR (for the non-reduced value of a), and so on.  Likewise,
the non-reduced (to underlying parameters) value of aA is aANR, and so on.
*)


(* Put in form of eqn A2: *)

laplaceMbzNumerator = p * Numerator[laplaceMbz]
laplaceMbzDenominator = p * Denominator[laplaceMbz]

Det[laplaceA]
```

$$\text{p}\left(\frac{\text{dwa dwb kb w1}^2\text{ zai}}{\text{p}\left(\text{dwa}^2+\text{w1}^2\right)}+\left(\text{dwb}^2+(\text{kb}+\text{p}+\text{r2b})^2\right)\left(\frac{\text{r1b}+\frac{\text{dwa}^2\text{ kb zai}}{\text{dwa}^2+\text{w1}^2}}{\text{p}}+\text{zbi}\right)\right)$$

$$\text{p}\left((\text{kb}+\text{p}+\text{r1b})\left(\text{dwb}^2+(\text{kb}+\text{p}+\text{r2b})^2\right)+(\text{kb}+\text{p}+\text{r2b})\text{ w1}^2\right)$$

$$(\text{kb}+\text{p}+\text{r1b})\left(\text{dwb}^2+(\text{kb}+\text{p}+\text{r2b})^2\right)+(\text{kb}+\text{p}+\text{r2b})\text{ w1}^2$$

```
(* Since lapaceMz can be represented by partial fractions form aA/(p+a) +
 (bB*(p+b)+cC)/((p+b)^2+s^2) + dD/p, can solve as in Morris & Chivers: *)

(* Hence, folling Morris eqn 25, eqn A6: *)
dD = Simplify[Expand[laplaceMbzNumerator ] /. p → 0] /
  ( (r1b + kb) * ((r2b + kb) ^2 + dwb ^2) + (r2b + kb) * w1 ^2 )
```

$$\left(r1b\left(dwb^2+(kb+r2b)^2\right)w1^2+dwa\,dwb\,kb\,w1^2\,zai+dwa^2\left(dwb^2+(kb+r2b)^2\right)(r1b+kb\,zai)\right)\Big/\left(\left(dwa^2+w1^2\right)\left((kb+r1b)\left(dwb^2+(kb+r2b)^2\right)+(kb+r2b)\,w1^2\right)\right)$$

```
(* Morris eqn 26, with aNR = a ("non-reduced") and gamma = a*( (b-a)^2 + s^2 ),
matching paper eqn 27 and from Morris, after their eqn 39 *)

aA = Collect[Simplify[(Expand[-laplaceMbzNumerator ] /. p → -aNR) / gamma],
  {zbi, zai}, Simplify]
```

$$-\frac{r1b\left(aNR^2+dwb^2-2\,aNR\,(kb+r2b)+(kb+r2b)^2\right)}{gamma}-\frac{dwa\,kb\left(aNR^2\,dwa-2\,aNR\,dwa\,(kb+r2b)+dwa\left(dwb^2+(kb+r2b)^2\right)+dwb\,w1^2\right)zai}{gamma\left(dwa^2+w1^2\right)}+\frac{aNR\left(aNR^2+dwb^2-2\,aNR\,(kb+r2b)+(kb+r2b)^2\right)zbi}{gamma}$$

```
(* Check that above matches what I have in paper in eqn 24:*)

Simplify[aA[[1]] - (-(r1b * ( (r2b + kb - aNR) ^2 + dwb ^2))) / gamma]
Simplify[aA[[2]] - (-zai * kb *
    (w1 ^2 * dwa * dwb + dwa ^2 * (dwb ^2 + (kb - aNR + r2b) ^2)) / (gamma * (dwa^2 + w1^2)))]
Simplify[aA[[3]] - (zbi * aNR * (dwb ^2 + (r2b + kb - aNR) ^2)) / gamma]
0

0

0
```

```
(* from Morris eqn 30,
bB= zbi*Mb0 - aA - dD .  Note that Morris is inconsistent on whether A,
B,C, and D include M0.  I'm including it. Note that
  I will use aANR to represent the non-reduced value of aA,
i.e. I want to refer to the aA value without reducing it
 to its parameter values given in eqn 24. *)

bB = Collect[Simplify[zbi - aANR - dD], {zbi, zai, aANR}, Simplify]
 (* note unlike Morris, zbi = Mbz/Mb0 is already normalized *)
```

$$-\mathrm{aANR}-\frac{\mathrm{r1b}\left(\mathrm{dwb}^2+(\mathrm{kb}+\mathrm{r2b})^2\right)}{\mathrm{dwb}^2\,(\mathrm{kb}+\mathrm{r1b})+(\mathrm{kb}+\mathrm{r2b})\left(\mathrm{kb}^2+\mathrm{r1b}\,\mathrm{r2b}+\mathrm{kb}\,(\mathrm{r1b}+\mathrm{r2b})+\mathrm{w1}^2\right)}-\frac{\mathrm{dwa}\,\mathrm{kb}\left(\mathrm{dwa}\left(\mathrm{dwb}^2+(\mathrm{kb}+\mathrm{r2b})^2\right)+\mathrm{dwb}\,\mathrm{w1}^2\right)\mathrm{zai}}{\left(\mathrm{dwa}^2+\mathrm{w1}^2\right)\left(\mathrm{dwb}^2\,(\mathrm{kb}+\mathrm{r1b})+(\mathrm{kb}+\mathrm{r2b})\left(\mathrm{kb}^2+\mathrm{r1b}\,\mathrm{r2b}+\mathrm{kb}\,(\mathrm{r1b}+\mathrm{r2b})+\mathrm{w1}^2\right)\right)}+\mathrm{zbi}$$

```
(* Check that above matches what I have in paper in eqn 25: *)
Simplify[
 bB - ( zbi - aANR - zai * dwa * kb * (w1 ^ 2 * dwb + dwa * (dwb ^ 2 + (r2b + kb) ^ 2)) /
      ((w1 ^ 2 + dwa ^ 2) * (w1 ^ 2 * (r2b + kb) + (r1b + kb) * (dwb ^ 2 + (r2b + kb) ^ 2))) -
    r1b * (dwb ^ 2 + (r2b + kb) ^ 2) / (w1 ^ 2 * (r2b + kb) +
         (r1b + kb) * (dwb ^ 2 + (r2b + kb) ^ 2)))]
```

0

In[20]:=
```
(* from Morris eqn 30, cC = d/dtau Mbz (t=0) + a*aA + b*bB *)
(* from lab notebook 27, p.25,  d/d tau Mbz (t=0) =
 -(alpha+k)*zbi*Mb0 + Mb0* (k*deltaA^2*zai/(1+deltaA^2) + alpha)  *)
(* cC = -(alpha+k)*zbi*Mb0 +
  Mb0* (k*deltaA^2*zai/(1+deltaA^2) + alpha) + aNR*aA + bNR*bB *)

(* redo using notation that is normalized by Mb0 and keeps w1 separate:*)
(* d/dt Mbz/Mb0(t=0) = (-r1b-kb)*Zbi + kb*dwa^2 * Zia/(w1^2+dwa^2) +r1b *)
(* cC = d Zb/dt (t=0) + aNR* aA + bNR * bB *)

(* Eqn 26: *)
 cC = Simplify[
  (-r1b - kb) * Zib + kb * dwa ^ 2 * Zia / (w1 ^ 2 + dwa ^ 2) + r1b + aNR * aANR + bNR * bBNR]
```

Out[20]= $\mathrm{aANR}\,\mathrm{aNR}+\mathrm{bBNR}\,\mathrm{bNR}+\mathrm{r1b}+\frac{\mathrm{dwa}^2\,\mathrm{kb}\,\mathrm{Zia}}{\mathrm{dwa}^2+\mathrm{w1}^2}-\mathrm{kb}\,\mathrm{Zib}-\mathrm{r1b}\,\mathrm{Zib}$